```
1   .define HEX_ADDRESS 0x2000
2   .define SW_ADDRESS 0x3000
3   .define MAX_SPEED 0x1000
4   .define STACK 255
5   .define COUNT_STORAGE 250
6
7   // This program displays a counter on the 7 segments
8   // The speed of the counter is controlled by the switches
9               mvi     r0, #0              // Used for counting
10              mvi     r1, #1              // Used for add/sub 1
11  MAIN:       mvi     r4, #SW_ADDRESS     // Point to switches
12              ld      r6, [r4]            // Read SW values
13              add     r6, r1              // Add 1 for minimum delay
14
15  // Count down delay until it reaches 0
16  DELAY1:     mvi     r5, #MAX_SPEED      // Reset max speed delay counter
17              mvi     r3, #DELAY2         // Point to inner delay loop
18
19  // Each delay counter will count MAX_SPEED times
20  DELAY2:     sub     r5, r1              // Count down by 1's
21              mvnz    r7, r3              // Continue inner delay loop
22  // End of DELAY2
23
24              sub     r6, r1              // Count down by 1's
25              mvi     r3, #DELAY1         // Point to outer delay loop
26              mvnz    r7, r3              // Continue outer delay loop
27  // End of DELAY1
28
29              add     r0, r1              // Increment counter
30              mvi     r3, #COUNT_STORAGE  // Store the counter because r0 is going to be
                modified
31              st      r0, [r3]
32
33  // Display the counter in decimal
34              mvi     r4, #HEX_ADDRESS    // Point to HEX port
35              mvi     r6, #6              // Used to count number of times DISPLAY must
                loop
36  DISPLAY:    mv      r5, r7              // Return address for DIV10
37              mvi     r7, #DIV10          // Call DIV10 subroutine
38
39              mvi     r5, #DATA           // Used to get display pattern
40
41              add     r5, r0              // Point to correct display pattern
42              ld      r3, [r5]            // Load display pattern
43              st      r3, [r4]            // Light up HEX display
44              add     r4, r1              // Go to next HEX display
45
46              mv      r0, r2              // Move quotient to r0 for next division
47
48              sub     r6, r1              // Decrement number of times DISPLAY still need
                to loop
49              mvi     r3, #DISPLAY        // Point to display loop
50              mvnz    r7, r3              // Keep looping until DISPLAY has looped 6
                times (r6=0)
51  // End of DISPLAY
52
53              mvi     r3, #COUNT_STORAGE  // Get the actual counter back
54              ld      r0, [r3]
55              mvi     r7, #MAIN           // Endless looping
56
57  // subroutine DIV10
58  // This subroutine divides the number in r0 by 10
59  // The algorithm subtracts 10 from r0 until r0 < 10, and keeps count in r2
60  // input: r0
61  // returns: quotient Q in r2, remainder R in r0
62  DIV10:      mvi     r1, #1
63              mvi     r3, #STACK          // Save registers on stack
64              sub     r3, r1              // save registers that are modified
65              st      r6, [r3]
```

```
66                  sub     r3, r1
67                  st      r4, [r3]            // end of register saving
68                  mvi     r2, #0              // init Q
69                  mvi     r6, RETDIV          // for branching
70
71  DLOOP:          mvi     r4, #9              // check if r0 is < 10 yet
72                  sub     r4, r0
73                  mvnc    r7, r6              // if so, then return
74
75  INC:            add     r2, r1              // but if not, then increment Q
76                  mvi     r4, #10
77                  sub     r0, r4              // r0 -= 10
78                  mvi     r7, DLOOP           // continue loop
79
80  RETDIV:         ld      r4, [r3]            // restore saved regs
81                  add     r3, r1
82                  ld      r6, [r3]            // restore the return address
83                  add     r3, r1
84                  add     r5, r1              // adjust the return address by 2
85                  add     r5, r1
86                  mv      r7, r5              // return results
87
88  // DATA for 7 segments
89  DATA:           .word 0b00111111            // '0'
90                  .word 0b00000110            // '1'
91                  .word 0b01011011            // '2'
92                  .word 0b01001111            // '3'
93                  .word 0b01100110            // '4'
94                  .word 0b01101101            // '5'
95                  .word 0b01111101            // '6'
96                  .word 0b00000111            // '7'
97                  .word 0b01111111            // '8'
98                  .word 0b01101111            // '9'
```